

An Introduction to PROC REPORT

Kirk Paul Lafler, Software Intelligence Corporation, Spring Valley, California

Abstract

SAS® users often need to create and deliver quality custom reports and specialized output for management, end users, and customers. The SAS System provides users with the REPORT procedure, a “canned” Base-SAS procedure, for producing quick and formatted detail and summary results. This presentation is designed for users who have no formal experience working with the REPORT procedure. Attendees learn the basic PROC REPORT syntax using the COLUMN, DEFINE, other optional statements, and procedure options to produce quality output; explore basic syntax to produce basic reports; compute subtotals and totals at the end of a report using a COMPUTE Block; calculate percentages; produce statistics for an analysis variable; apply conditional logic to control summary output rows; and enhance the appearance of output results with basic Output Delivery System (ODS) techniques.

Introduction

The creation of any good report begins by identifying the specific needs of the target audience. This typically requires attention be given to the report’s structure, its organization and finally to the exact information that is intended to be conveyed. Reporting in the SAS System, as with any software, requires data to first be collected, then organized and finally presented in a logical and concise way. SAS users have many ways to create quality reports and output, including the use of DATA _NULL_ reporting techniques and an assortment of detail, summary, statistical, frequency, graphical and tabular procedures to choose from. But one method is traditionally chosen by today’s SAS user more than any other approach – the selection of the REPORT procedure which just happens to be part of the SAS Base software. This paper introduces SAS users to the basic features and capabilities of the REPORT procedure in the creation of simple detail and summary reports and output.

Data Set Used in Examples

The examples used throughout this paper utilize a data set called, MOVIES. The Movies table consists of six columns: title, length, category, year, studio, and rating. Title, category, studio, and rating are defined as character columns with length and year being defined as numeric columns, shown below.

	Title	Length	Category	Year	Studio	Rating
1	Brave Heart	177	Action Adventure	1995	Paramount Pictures	R
2	Casablanca	103	Drama	1942	MGM / UA	PG
3	Christmas Vacation	97	Comedy	1989	Warner Brothers	PG-13
4	Coming to America	116	Comedy	1988	Paramount Pictures	R
5	Dracula	130	Horror	1993	Columbia TriStar	R
6	Dressed to Kill	105	Drama Mysteries	1980	Filmways Pictures	R
7	Forrest Gump	142	Drama	1994	Paramount Pictures	PG-13
8	Ghost	127	Drama Romance	1990	Paramount Pictures	PG-13
9	Jaws	125	Action Adventure	1975	Universal Studios	PG
10	Jurassic Park	127	Action	1993	Universal Pictures	PG-13
11	Lethal Weapon	110	Action Cops & Robber	1987	Warner Brothers	R
12	Michael	106	Drama	1997	Warner Brothers	PG-13
13	National Lampoon's Vacation	98	Comedy	1983	Warner Brothers	PG-13
14	Poltergeist	115	Horror	1982	MGM / UA	PG
15	Rocky	120	Action Adventure	1976	MGM / UA	PG
16	Scarface	170	Action Cops & Robber	1983	Universal Studios	R
17	Silence of the Lambs	118	Drama Suspense	1991	Orion	R
18	Star Wars	124	Action Sci-Fi	1977	Lucas Film Ltd	PG
19	The Hunt for Red October	135	Action Adventure	1989	Paramount Pictures	PG
20	The Terminator	108	Action Sci-Fi	1984	Live Entertainment	R
21	The Wizard of Oz	101	Adventure	1939	MGM / UA	G
22	Titanic	194	Drama Romance	1997	Paramount Pictures	PG-13

Basic PROC REPORT Statement Syntax

The REPORT procedure is a powerful tool for creating detail and summary reports and output. As with all powerful procedures, users have a number of statements and options to choose from. In its simplest form, below, PROC REPORT with the DATA= option displays all variables for all observations in the specified SAS data set. SAS determines the best way to format the output, so you don't necessarily need to worry about these types of issues. It should be noted that the order of the displayed variables on the output report is not alphabetical, but the order the variables were created in the data set. It should also be noted that when a Title statement is not specified, the default title displayed on output is, "The SAS System." The basic syntax of PROC REPORT with the DATA= option and the corresponding output is shown below.

PROC REPORT Code

```
PROC REPORT DATA=mydata.movies ;
RUN ;
```

Output

The SAS System					
Title	Length	Category	Year	Studio	Rating
Brave Heart	177	Action Adventure	1995	Paramount Pictures	R
Casablanca	103	Drama	1942	MGM / UA	PG
Christmas Vacation	97	Comedy	1989	Warner Brothers	PG-13
Coming to America	116	Comedy	1988	Paramount Pictures	R
Dracula	130	Horror	1993	Columbia TriStar	R
Dressed to Kill	105	Drama Mysteries	1980	Filmways Pictures	R
Forrest Gump	142	Drama	1994	Paramount Pictures	PG-13
Ghost	127	Drama Romance	1990	Paramount Pictures	PG-13
Jaws	125	Action Adventure	1975	Universal Studios	PG
Jurassic Park	127	Action	1993	Universal Pictures	PG-13
Lethal Weapon	110	Action Cops & Robber	1987	Warner Brothers	R
Michael	106	Drama	1997	Warner Brothers	PG-13
National Lampoon's Vacation	98	Comedy	1983	Warner Brothers	PG-13
Poltergeist	115	Horror	1982	MGM / UA	PG
Rocky	120	Action Adventure	1976	MGM / UA	PG
Scarface	170	Action Cops & Robber	1983	Universal Studios	R
Silence of the Lambs	118	Drama Suspense	1991	Orion	R
Star Wars	124	Action Sci-Fi	1977	Lucas Film Ltd	PG
The Hunt for Red October	135	Action Adventure	1989	Paramount Pictures	PG
The Terminator	108	Action Sci-Fi	1984	Live Entertainment	R
The Wizard of Oz	101	Adventure	1939	MGM / UA	G
Titanic	194	Drama Romance	1997	Paramount Pictures	PG-13

Users of the REPORT procedure have even greater control over input, processing and output scenarios, with an assortment of options. The table, illustrated below, shows several PROC REPORT options (in alphabetical order) along with a brief description of each.

Option	Description
DATA=	Specify the input data set to be used for processing.
HEADLINE	Create a horizontal line between the column headers and the body of the report.
HEADSKIP	Create a blank line between the column headers and the body of the report.
NOWINDOWS	Suppress the REPORT window and direct the report output to open ODS destinations.
OUT=	Create an output SAS data set.
OUTREPT=	Specify a location to store the report.
PROMPT	Activate prompting mode by using a built-in wizard.
REPORT=	Specify a stored report to be used in generating a new report.

Selecting Variables to Display with a KEEP= Data Set Option

PROC REPORT output can also be produced with a KEEP= data set option. In lieu of all the variables being read from the input SAS data set, a KEEP= data set option gives users a controlled way to select the variable(s) needed from an input data set. This not only prevents all variables from being read and processed, it reduces CPU and I/O performance demands on the system. As was presented in the previous example, the order of the displayed variables for the generated output is in the order the variables were created in the data set. The following example shows the syntax and corresponding output of a KEEP= data set and the NOWINDOWS option.

PROC REPORT Code

```
PROC REPORT DATA=mydata.movies(KEEP=title rating category length) NOWINDOWS ;
RUN ;
```

Output

Title	Length	Category	Rating
Brave Heart	177	Action Adventure	R
Casablanca	103	Drama	PG
Christmas Vacation	97	Comedy	PG-13
Coming to America	116	Comedy	R
Dracula	130	Horror	R
Dressed to Kill	105	Drama Mysteries	R
Forrest Gump	142	Drama	PG-13
Ghost	127	Drama Romance	PG-13
Jaws	125	Action Adventure	PG
Jurassic Park	127	Action	PG-13
Lethal Weapon	110	Action Cops & Robber	R
Michael	106	Drama	PG-13
National Lampoon's Vacation	98	Comedy	PG-13
Poltergeist	115	Horror	PG
Rocky	120	Action Adventure	PG
Scarface	170	Action Cops & Robber	R
Silence of the Lambs	118	Drama Suspense	R
Star Wars	124	Action Sci-Fi	PG
The Hunt for Red October	135	Action Adventure	PG
The Terminator	108	Action Sci-Fi	R
The Wizard of Oz	101	Adventure	G
Titanic	194	Drama Romance	PG-13

PROC REPORT Statements

The REPORT procedure provides users with a number of optional statements to help gain greater control over the processing and layout of report output.

PROC REPORT Code

```
PROC REPORT data=SAS-data-set options ;
  COLUMNS variable_1 - variable_n ;
  DEFINE variable_1 / options ;
  DEFINE variable_2 / options ;
  . . . . .
  DEFINE variable_n / options ;
  COMPUTE blocks ;   ENDCOMPUTE ;
  BREAK ;
  RBREAK ;
RUN ;
```

The following table lists the various optional PROC REPORT statements along with a brief description.

Statement	Description
COLUMNS	Specify the variable(s) to appear in the report and their order.
DEFINE	Define how each variable will be used in the report.
COMPUTE / ENDCOMPUTE	Define the calculations to be performed in the report.
BREAK	Produce a break or summary within a report either before or after a change in the value of a grouping (or by-group) variable.
RBREAK	Produce a summary across the entire report (report wide) and can be placed at the beginning or end of a report.

Selecting Variables to Display with a COLUMNS Statement

PROC REPORT output can also be produced with an optional COLUMNS statement. The COLUMNS statement tells SAS what variable(s) to display in the report. Unlike the previous examples, the variables specified with the COLUMNS statement are displayed in the exact order specified and not in the order the variables were created in the data set. The following example shows the syntax and corresponding output of the COLUMNS statement.

PROC REPORT Code

```
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS title rating category length ;
RUN ;
```

Output

The SAS System

Title	Rating	Category	Length
Brave Heart	R	Action Adventure	177
Casablanca	PG	Drama	103
Christmas Vacation	PG-13	Comedy	97
Coming to America	R	Comedy	116
Dracula	R	Horror	130
Dressed to Kill	R	Drama Mysteries	105
Forrest Gump	PG-13	Drama	142
Ghost	PG-13	Drama Romance	127
Jaws	PG	Action Adventure	125
Jurassic Park	PG-13	Action	127
Lethal Weapon	R	Action Cops & Robber	110
Michael	PG-13	Drama	106
National Lampoon's Vacation	PG-13	Comedy	98
Pokey	PG	Horror	115
Rocky	PG	Action Adventure	120
Scarface	R	Action Cops & Robber	170
Silence of the Lambs	R	Drama Suspense	118
Star Wars	PG	Action Sci-Fi	124
The Hunt for Red October	PG	Action Adventure	135
The Terminator	R	Action Sci-Fi	108
The Wizard of Oz	G	Adventure	101
Titanic	PG-13	Drama Romance	194

Defining How Variables are Used with a DEFINE Statement and Options

PROC REPORT output can also be produced with an optional DEFINE statement and options. The DEFINE statement tells PROC REPORT how each variable will be used in the generated report output. The following table illustrates the various DEFINE statement options along with a brief description.

Option	Description
ACROSS	Defines the item on the report as an across variable.
ANALYSIS	Define the item on the report as an analysis variable.
CENTER	Center the column headers and formatted values.
COMPUTED	Define the item on the report as a computed variable.
DISPLAY	Define the item on the report as a display variable.
FORMAT=	Assign a SAS or user-defined format to the item.
GROUP	Display the item on the report as a group variable (categories).
MISSING	Display missing values as valid values for the item.
ORDER	Display the item on the report as an order variable.
WIDTH=	Define the width of the column to display on the report.

In the next example, the DEFINE statement specifies the DISPLAY option with a character string to use as the variable heading for each variable. Each DEFINE statement also tells PROC REPORT the width to display each column's content using the WIDTH= option. Unlike the COLUMNS statement specified in the previous example, the DEFINE statement's purpose is to tell PROC REPORT how each variable is to be used and displayed in the report. The resulting report output displays all variables in the order they were created in the data set, shown below.

PROC REPORT Code

```
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / DISPLAY 'Movie Rating' WIDTH=5 ;
  DEFINE category / DISPLAY 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

The SAS System					
Movie Title	Movie Length	Category	Year	Studio	Movie Rating
Brave Heart	177	Action Adventure	1995	Paramount Pictures	R
Casablanca	103	Drama	1942	MGM / UA	PG
Christmas Vacation	97	Comedy	1989	Warner Brothers	PG-13
Coming to America	116	Comedy	1988	Paramount Pictures	R
Dracula	130	Horror	1993	Columbia TriStar	R
Dressed to Kill	105	Drama Mysteries	1980	Filmways Pictures	R
Forrest Gump	142	Drama	1994	Paramount Pictures	PG-13
Ghost	127	Drama Romance	1990	Paramount Pictures	PG-13
Jaws	125	Action Adventure	1975	Universal Studios	PG
Jurassic Park	127	Action	1993	Universal Pictures	PG-13
Lethal Weapon	110	Action Cops & Robber	1987	Warner Brothers	R
Michael	108	Drama	1997	Warner Brothers	PG-13
National Lampoon's Vacation	98	Comedy	1983	Warner Brothers	PG-13
Poltergeist	115	Horror	1982	MGM / UA	PG
Rocky	120	Action Adventure	1976	MGM / UA	PG
Scarface	170	Action Cops & Robber	1983	Universal Studios	R
Silence of the Lambs	118	Drama Suspense	1991	Orion	R
Star Wars	124	Action Sci-Fi	1977	Lucas Film Ltd	PG
The Hunt for Red October	135	Action Adventure	1989	Paramount Pictures	PG
The Terminator	108	Action Sci-Fi	1984	Live Entertainment	R
The Wizard of Oz	101	Adventure	1939	MGM / UA	G
Titanic	194	Drama Romance	1997	Paramount Pictures	PG-13

Creating PROC REPORT Output with a COLUMNS and DEFINE Statement

The COLUMNS and DEFINE statements can also be combined in PROC REPORT to create an output report. The COLUMNS statement tells SAS what variable(s) to display in the report. It is also worth noting that the COLUMNS statement limits the display to only those columns specified. The DEFINE statement tells PROC REPORT how each variable will be used in the generated report output. The following example illustrates using a COLUMNS and DEFINE statement along with various display options.

PROC REPORT Code

```
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS title rating category length ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / DISPLAY 'Movie Rating' WIDTH=5 ;
  DEFINE category / DISPLAY 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

The SAS System

Movie Title	Movie Rating	Category	Movie Length
Brave Heart	R	Action Adventure	177
Casablanca	PG	Drama	103
Christmas Vacation	PG-13	Comedy	97
Coming to America	R	Comedy	116
Dracula	R	Horror	130
Dressed to Kill	R	Drama Mysteries	105
Forrest Gump	PG-13	Drama	142
Ghost	PG-13	Drama Romance	127
Jaws	PG	Action Adventure	125
Jurassic Park	PG-13	Action	127
Lethal Weapon	R	Action Cops & Robber	110
Michael	PG-13	Drama	106
National Lampoon's Vacation	PG-13	Comedy	98
Poltergeist	PG	Horror	115
Rocky	PG	Action Adventure	120
Scarface	R	Action Cops & Robber	170
Silence of the Lambs	R	Drama Suspense	118
Star Wars	PG	Action Sci-Fi	124
The Hunt for Red October	PG	Action Adventure	135
The Terminator	R	Action Sci-Fi	108
The Wizard of Oz	G	Adventure	101
Titanic	PG-13	Drama Romance	194

Displaying Titles on PROC REPORT Output

Adding Titles to PROC REPORT output is simple with SAS. One or more TITLE statements can be specified at the top of each page of output (a maximum of ten can be specified). It's worth noting that footnotes can be added at the bottom of each page of output with the FOOTNOTE statement (a maximum of ten can be specified). As in the previous example, the COLUMNS and DEFINE statements were specified in PROC REPORT to create an output report. The next example illustrates using a TITLE statement to add a title at the top of each page of output.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS title rating category length ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / DISPLAY 'Movie Rating' WIDTH=5 ;
  DEFINE category / DISPLAY 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

Detailed Movies Listing

Movie Title	Movie Rating	Category	Movie Length
Brave Heart	R	Action Adventure	177
Casablanca	PG	Drama	103
Christmas Vacation	PG-13	Comedy	97
Coming to America	R	Comedy	116
Dracula	R	Horror	130
Dressed to Kill	R	Drama Mysteries	105
Forrest Gump	PG-13	Drama	142
Ghost	PG-13	Drama Romance	127
Jaws	PG	Action Adventure	125
Jurassic Park	PG-13	Action	127
Lethal Weapon	R	Action Cops & Robber	110
Michael	PG-13	Drama	106
National Lampoon's Vacation	PG-13	Comedy	98
Polltergeist	PG	Horror	115
Rocky	PG	Action Adventure	120
Scarface	R	Action Cops & Robber	170
Silence of the Lambs	R	Drama Suspense	118
Star Wars	PG	Action Sci-Fi	124
The Hunt for Red October	PG	Action Adventure	135
The Terminator	R	Action Sci-Fi	108
The Wizard of Oz	G	Adventure	101
Titanic	PG-13	Drama Romance	194

Centering Data in a Column of PROC REPORT Output

The data displayed in each column of output can be centered. By default, PROC REPORT left justifies character-defined data and right justifies numeric-defined data. As in the previous example, the TITLE, COLUMNS and DEFINE statements were specified in PROC REPORT to create a “custom” output report. The next example illustrates using the DEFINE statement CENTER option to center the data for the RATING column on the report output.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS title rating category length ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / DISPLAY 'Movie Rating' WIDTH=5
                CENTER ;
  DEFINE category / DISPLAY 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

Detailed Movies Listing

Movie Title	Movie Rating	Category	Movie Length
Brave Heart	R	Action Adventure	177
Casablanca	PG	Drama	103
Christmas Vacation	PG-13	Comedy	97
Coming to America	R	Comedy	116
Dracula	R	Horror	130
Dressed to Kill	R	Drama Mysteries	105
Forrest Gump	PG-13	Drama	142
Ghost	PG-13	Drama Romance	127
Jaws	PG	Action Adventure	125
Jurassic Park	PG-13	Action	127
Lethal Weapon	R	Action Cops & Robber	110
Michael	PG-13	Drama	106
National Lampoon's Vacation	PG-13	Comedy	98
Poltergeist	PG	Horror	115
Rocky	PG	Action Adventure	120
Scarface	R	Action Cops & Robber	170
Silence of the Lambs	R	Drama Suspense	118
Star Wars	PG	Action Sci-Fi	124
The Hunt for Red October	PG	Action Adventure	135
The Terminator	R	Action Sci-Fi	108
The Wizard of Oz	G	Adventure	101
Titanic	PG-13	Drama Romance	194

Creating a Grouped PROC REPORT Output

A “grouped” type of output can be created with PROC REPORT where analysis variables can be summarized based on a grouping variable. The next example illustrates a COLUMNS statement containing the desired variables to display on the report output. Notice that the order of the TITLE and RATING variables specified in the COLUMNS statement have been switched where now the categorical variable, RATING, is specified first. The DEFINE statement for the RATING variable specifies a GROUP option to create a “grouped” type of output where all the observations in the GROUP variable are consolidated together.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
TITLE2 Grouped by Rating ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS rating title category length ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / GROUP 'Movie Rating' WIDTH=5
                CENTER ;
  DEFINE category / DISPLAY 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

Detailed Movies Listing
Grouped by Rating

Movie Rating	Movie Title	Category	Movie Length
G	The Wizard of Oz	Adventure	101
PG	Casablanca	Drama	103
	Jaws	Action Adventure	125
	Polltergeist	Horror	115
	Rocky	Action Adventure	120
	Star Wars	Action Sci-Fi	124
	The Hunt for Red October	Action Adventure	135
PG-13	Christmas Vacation	Comedy	97
	Forrest Gump	Drama	142
	Ghost	Drama Romance	127
	Jurassic Park	Action	127
	Michael	Drama	106
	National Lampoon's Vacation	Comedy	98
R	Titanic	Drama Romance	194
	Brave Heart	Action Adventure	177
	Coming to America	Comedy	116
	Dracula	Horror	130
	Dressed to Kill	Drama Mysteries	105
	Lethal Weapon	Action Cops & Robber	110
	Scarface	Action Cops & Robber	170
	Silence of the Lambs	Drama Suspense	118
The Terminator	Action Sci-Fi	108	

After running the preceding examples' PROC REPORT code, you may receive the SAS Log message:

NOTE: Groups are not created because the usage of Title is DISPLAY. To avoid this note, change all GROUP variables to ORDER variables.

To prevent this message from being generated on the SAS Log, the next example changes the DEFINE statement for the RATING variable from a "GROUP" option to an "ORDER" option, as shown below. The Log message is removed.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
TITLE2 Sorted by Rating ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS rating title category length ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / ORDER 'Movie Rating' WIDTH=5
                CENTER ;
  DEFINE category / DISPLAY 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

Movie Rating	Movie Title	Category	Movie Length
G	The Wizard of Oz	Adventure	101
PG	Casablanca	Drama	103
	Jaws	Action Adventure	125
	Pollergelst	Horror	115
	Rocky	Action Adventure	120
	Star Wars	Action Sci-Fi	124
	The Hunt for Red October	Action Adventure	135
PG-13	Christmas Vacation	Comedy	97
	Forrest Gump	Drama	142
	Ghost	Drama Romance	127
	Jurassic Park	Action	127
	Michael	Drama	106
	National Lampoon's Vacation	Comedy	98
	Titanic	Drama Romance	194
R	Brave Heart	Action Adventure	177
	Coming to America	Comedy	116
	Dracula	Horror	130
	Dressed to Kill	Drama Mysteries	105
	Lethal Weapon	Action Cops & Robber	110
	Scarface	Action Cops & Robber	170
	Silence of the Lambs	Drama Suspense	118
	The Terminator	Action Sci-Fi	108

The next example illustrates another “grouped” report output illustrates the COLUMNS statement specifying the categorical variable, CATEGORY, is specified first. The DEFINE statement for the CATEGORY variable specifies an ORDER option to create an “ordered” type of output where all the observations in the ORDER variable are consolidated together.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
TITLE2 Sorted by Category ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS category rating title length ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / DISPLAY 'Movie Rating' WIDTH=5
              CENTER ;
  DEFINE category / ORDER 'Category' WIDTH=20 ;
  DEFINE length / DISPLAY 'Movie Length' WIDTH=3 ;
RUN ;
```

Output

Detailed Movies Listing
Sorted by Category

Category	Movie Rating	Movie Title	Movie Length
Action	PG-13	Jurassic Park	127
Action Adventure	R	Brave Heart	177
	PG	Jaws	125
	PG	Rocky	120
	PG	The Hunt for Red October	135
Action Cops & Robber	R	Lethal Weapon	110
	R	Scarface	170
Action Sci-Fi	PG	Star Wars	124
	R	The Terminator	108
Adventure	G	The Wizard of Oz	101
Comedy	PG-13	Christmas Vacation	97
	R	Coming to America	116
	PG-13	National Lampoon's Vacation	98
Drama	PG	Casablanca	103
	PG-13	Forrest Gump	142
	PG-13	Michael	106
Drama Mysteries	R	Dressed to Kill	105
Drama Romance	PG-13	Ghost	127
	PG-13	Titanic	194
Drama Suspense	R	Silence of the Lambs	118
Horror	R	Dracula	130
	PG	Pottergeist	115

Calculating Subtotals, Percentages and Statistics Output

The next example illustrates how PROC REPORT computes and displays a new variable on the report output. As before, the COLUMNS statement selects the variables, including the computed variable, LENGTH2. The DEFINE statement identifies LENGTH as an ANALYSIS variable and another DEFINE statement identifies LENGTH2 as a COMPUTED variable. Finally, the COMPUTE / ENDCOMP block performs the computation using the SUM function.

PROC REPORT Code

```

TITLE Detailed Movies Listing ;
TITLE2 Sorted by Category ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
    COLUMNS category rating title length length2 ;
    DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
    DEFINE rating / DISPLAY 'Movie Rating' WIDTH=5 CENTER ;
    DEFINE category / ORDER 'Category' WIDTH=20 ;
    DEFINE length / ANALYSIS 'Movie Length' WIDTH=3 ;
    DEFINE length2 / COMPUTED 'Trailer Length' ;
    COMPUTE length2 ;
        length2 = 1 + length.sum ;
    ENDCOMP ;
RUN ;
    
```

Output

Detailed Movies Listing
Sorted by Category

Category	Movie Rating	Movie Title	Movie Length	Trailer Length
Action	PG-13	Jurassic Park	127	128
Action Adventure	R	Brave Heart	177	178
	PG	Jaws	125	126
	PG	Rocky	120	121
	PG	The Hunt for Red October	135	136
Action Cops & Robber	R	Lethal Weapon	110	111
	R	Scarface	170	171
Action Sci-Fi	PG	Star Wars	124	125
	R	The Terminator	108	109
Adventure	G	The Wizard of Oz	101	102
Comedy	PG-13	Christmas Vacation	97	98
	R	Coming to America	116	117
	PG-13	National Lampoon's Vacation	98	99
Drama	PG	Casablanca	103	104
	PG-13	Forrest Gump	142	143
	PG-13	Michael	106	107
Drama Mysteries	R	Dressed to Kill	105	106
Drama Romance	PG-13	Ghost	127	128
	PG-13	Titanic	194	195
Drama Suspense	R	Silence of the Lambs	118	119
Horror	R	Dracula	130	131
	PG	Poltergeist	115	116

Applying Conditional Logic in PROC REPORT Code

The next example illustrates how conditional logic can be optionally applied in PROC REPORT code to control the display of information on report output and, specifically the display of computed values for the variable, LENGTH2.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
TITLE2 Using Conditional Logic ;
PROC REPORT DATA=mydata.movies NOWINDOWS ;
  COLUMNS rating title studio length length2 ;
  DEFINE title / DISPLAY 'Movie Title' WIDTH=30 ;
  DEFINE rating / ORDER 'Movie Rating' WIDTH=5 CENTER ;
  DEFINE studio / DISPLAY 'Movie Studio' WIDTH=20 ;
  DEFINE length / ANALYSIS 'Movie Length' WIDTH=3 ;
  DEFINE length2 / COMPUTED 'Trailer Length' ;
  COMPUTE length2 ;
    if UPCASE(studio)='PARAMOUNT PICTURES' then
      length2 = 1 + length.sum ;
  ENDCOMP ;
RUN ;
```

Output

Detailed Movies Listing
Using Conditional Logic

Movie Rating	Movie Title	Movie Studio	Movie Length	Trailer Length
G	The Wizard of Oz	MGM / UA	101	.
PG	Casablanca	MGM / UA	103	.
	Jaws	Universal Studios	125	.
	Polltergeist	MGM / UA	115	.
	Rocky	MGM / UA	120	.
	Star Wars	Lucas Film Ltd	124	.
	The Hunt for Red October	Paramount Pictures	135	136
PG-13	Christmas Vacation	Warner Brothers	97	.
	Forrest Gump	Paramount Pictures	142	143
	Ghost	Paramount Pictures	127	128
	Jurassic Park	Universal Pictures	127	.
	Michael	Warner Brothers	106	.
	National Lampoon's Vacation	Warner Brothers	98	.
	Titanic	Paramount Pictures	194	195
R	Brave Heart	Paramount Pictures	177	178
	Coming to America	Paramount Pictures	116	117
	Dracula	Columbia TriStar	130	.
	Dressed to Kill	Filmways Pictures	105	.
	Lethal Weapon	Warner Brothers	110	.
	Scarface	Universal Studios	170	.
	Silence of the Lambs	Orion	118	.
	The Terminator	Live Entertainment	108	.

Enhancing the Appearance of PROC REPORT Output

The appearance of PROC REPORT output can be enhanced by specifying one or more components and/or one or more style attributes on the PROC REPORT statement. The following tables describe available component options and style attributes that users can specify to control various parts of a report's output, see below.

Component	Description
Report	Affects the report and the table structure.
Header (HDR)	Affects color header cells.
Column	Affects data cells.
Summary	Affects summary lines generated by BREAK or RBREAK statements.
Lines	Affects lines generated by LINE statements.

Style Attribute	Description
Background=	Changes background color (e.g., Red, Blue, Green, Orange, Yellow).
Bordercolor=	Changes border color (e.g., Red, Blue, Green, Orange, Yellow).
Borderwidth=	Changes border width (e.g., 0, 3, 7, in, cm, pt).
Cellspacing=	Changes cell spacing (e.g., 0, 3, 7, in, cm, pt).
Foreground=	Changes foreground color (e.g., Red, Blue, Green, Orange, Yellow).
Frame=	Changes frame style (e.g., ABOVE, BELOW, BOX, HSIDES, LHS, RHS, VOID and VSIDES).
Rules=	Changes rules (e.g., ALL, COLS, GROUPS, NONE, and ROWS).

In the next PROC REPORT code example, a **STYLE(Column)** component option is specified to instruct SAS to use the color "Blue" for the background and "White" for the foreground of each cell on the report output.

PROC REPORT Code

```
TITLE Detailed Movies Listing ;
TITLE2 Using STYLE(Column)= ;
PROC REPORT DATA=mydata.movies NOWINDOWS
  STYLE(Column)=[Background=Blue Foreground=White Cellspacing=10] ;
  COLUMNS rating title studio length length2 ;
  DEFINE title / DISPLAY "Movie Title" WIDTH=30 ;
  DEFINE rating / ORDER "Movie Rating" WIDTH=5
                CENTER ;
  DEFINE studio / DISPLAY "Movie Studio" WIDTH=20 ;
  DEFINE length / ANALYSIS "Movie Length" WIDTH=3 ;
  DEFINE length2 / COMPUTED "Trailer Length" ;
  COMPUTE length2 ;
    if UPCASE(studio)="PARAMOUNT PICTURES" then
      length2 = 1 + length.sum ;
  ENDCOMP ;
RUN ;
```

Output

Detailed Movies Listing Using STYLE(Column)=				
Movie Rating	Movie Title	Movie Studio	Movie Length	Trailer Length
G	The Wizard of Oz	MGM / UA	101	-
PG	Casablanca	MGM / UA	103	-
	Jaws	Universal Studios	125	-
	Pollergeist	MGM / UA	115	-
	Rocky	MGM / UA	120	-
	Star Wars	Lucas Film Ltd	124	-
	The Hunt for Red October	Paramount Pictures	135	136
PG-13	Christmas Vacation	Warner Brothers	97	-
	Forrest Gump	Paramount Pictures	142	143
	Ghost	Paramount Pictures	127	128
	Jurassic Park	Universal Pictures	127	-
	Michael	Warner Brothers	106	-
	National Lampoon's Vacation	Warner Brothers	98	-
	Titanic	Paramount Pictures	194	195
R	Brave Heart	Paramount Pictures	177	178
	Coming to America	Paramount Pictures	116	117
	Dracula	Columbia TriStar	130	-
	Dressed to Kill	Filmways Pictures	105	-
	Lethal Weapon	Warner Brothers	110	-
	Scarface	Universal Studios	170	-
	Silence of the Lambs	Orion	118	-
	The Terminator	Live Entertainment	108	-

In the next PROC REPORT code example, a STYLE(Header) component option is specified to instruct SAS to use the color WHITE for the background and BLUE for the border on the report output.

PROC REPORT Code

```

TITLE Detailed Movies Listing ;
TITLE2 Using STYLE(Header)=;
PROC REPORT DATA=mydata.movies NOWINDOWS
      STYLE(Header)=[Background=White Cellspacing=0
                    Bordercolor=Blue Borderwidth=2
                    Rules=rows      Frame=box] ;
COLUMNS rating title studio length length2 ;
DEFINE title / DISPLAY "Movie Title" WIDTH=30 ;
DEFINE rating / ORDER "Movie Rating" WIDTH=5 CENTER ;
DEFINE studio / DISPLAY "Movie Studio" WIDTH=20 ;
DEFINE length / ANALYSIS "Movie Length" WIDTH=3 ;
DEFINE length2 / COMPUTED "Trailer Length" ;
COMPUTE length2 ;
      if UPCASE(studio)="PARAMOUNT PICTURES" then
          length2 = 1 + length.sum ;
ENDCOMP ;
RUN ;

```



Output

**Detailed Movies Listing
Using STYLE(Header)=**

Movie Rating	Movie Title	Movie Studio	Movie Length	Trailer Length
G	The Wizard of Oz	MGM / UA	101	.
PG	Casablanca	MGM / UA	103	.
	Jaws	Universal Studios	125	.
	Poltergeist	MGM / UA	115	.
	Rocky	MGM / UA	120	.
	Star Wars	Lucas Film Ltd	124	.
	The Hunt for Red October	Paramount Pictures	135	136
PG-13	Christmas Vacation	Warner Brothers	97	.
	Forrest Gump	Paramount Pictures	142	143
	Ghost	Paramount Pictures	127	128
	Jurassic Park	Universal Pictures	127	.
	Michael	Warner Brothers	106	.
	National Lampoon's Vacation	Warner Brothers	98	.
	Titanic	Paramount Pictures	194	195
R	Brave Heart	Paramount Pictures	177	178
	Coming to America	Paramount Pictures	116	117
	Dracula	Columbia TriStar	130	.
	Dressed to Kill	Filmways Pictures	105	.
	Lethal Weapon	Warner Brothers	110	.
	Scarface	Universal Studios	170	.
	Silence of the Lambs	Orion	118	.
	The Terminator	Live Entertainment	108	.

Creating Unique Output with ODS

Output Delivery System (ODS) can be used to create a variety of output formats. ODS statements are classified as global statements and are processed immediately by the SAS System. ODS statement options control what format engine(s) are turned on and in effect during the step or until another ODS statement is specified. ODS has built-in format engines (e.g., Listing, Output, RTF, PDF, DATA Step, HTML, Excel, PowerPoint and XML). Specifying an ODS statement and destination at a particular point in a program is important, because output-producing PROC and DATA steps will respond by sending output to the open destination.



Available “Custom” Styles

Although the default styles that SAS uses with their output look fine, users have many more choices to choose from. To view the available template styles, users can submit a LIST Styles statement using PROC TEMPLATE, below.

```

PROC TEMPLATE ;
  LIST STYLES ;
RUN ;
  
```

Output and SAS Styles

Obs	Path	Type
1	Styles	Dir
2	Styles.Analysis	Style
3	Styles.BarrettsBlue	Style
4	Styles.BrockPrint	Style
5	Styles.DTree	Style
6	Styles.Daisy	Style
7	Styles.Default	Style
8	Styles.Dove	Style
9	Styles.EGDefault	Style
10	Styles.Excel	Style
11	Styles.FancyPrinter	Style
12	Styles.Festival	Style
13	Styles.FestivalPrinter	Style
14	Styles.Gantt	Style
15	Styles.GrayscalePrinter	Style
16	Styles.HTMLBlue	Style
17	Styles.Harvest	Style
18	Styles.HighContrast	Style
19	Styles.HighContrastLarge	Style
20	Styles.Journal	Style
21	Styles.Journal1a	Style
22	Styles.Journal2	Style
23	Styles.Journal2a	Style
24	Styles.Journal3	Style
25	Styles.Journal3a	Style
26	Styles.Listing	Style
27	Styles.Meadow	Style
28	Styles.MeadowPrinter	Style
29	Styles.Minimal	Style
30	Styles.MonochromePrinter	Style
31	Styles.Monospace	Style
32	Styles.Moonflower	Style
33	Styles.Netdraw	Style
34	Styles.NoFontDefault	Style
35	Styles.Normal	Style
36	Styles.NormalPrinter	Style
37	Styles.Ocean	Style
38	Styles.Pearl	Style
39	Styles.PearlJ	Style
40	Styles.Plateau	Style
41	Styles.PowerPointDark	Style
42	Styles.PowerPointLight	Style
43	Styles.Printer	Style
44	Styles.Raven	Style
45	Styles.Rtf	Style
46	Styles.Sapphire	Style
47	Styles.SasDocPrinter	Style
48	Styles.SasWeb	Style
49	Styles.Seaside	Style
50	Styles.SeasidePrinter	Style
51	Styles.Snow	Style
52	Styles.StatDoc	Style
53	Styles.Statistical	Style
54	Styles.Word	Style
55	Styles.vaDark	Style
56	Styles.vaHighContrast	Style
57	Styles.vaLight	Style

The ODS Excel Destination

The **ODS Excel** Destination became production in SAS 9.4 (M3). It serves as an interface between SAS and Excel, and provides users with the following features:

- ✓ SAS Results and Output can be sent directly to Excel
- ✓ Offers a Flexible way to create Excel files
- ✓ Supports Reports, Tables, Statistics and Graphs
- ✓ Formats Data into Excel Worksheet cells
- ✓ Permits Automation of Production-level Workbooks

The ODS Excel destination easily sends output and results to Excel. The ODS Excel syntax simplifies the process of sending output, reports, tables, statistics and graphs to Excel files. The ODS Excel options are able to:

- ✓ Programmatically generate output and results
- ✓ Control font sizes
- ✓ Add special features to row and column headers
- ✓ Adjust row and column sizes
- ✓ Format data values
- ✓ Align data to the left, center or right
- ✓ Add hyperlinks for drill-down capability

In this next example an ODS Excel statement and a `STYLE=BarrettsBlue` option with PROC REPORT is specified to send the contents of the MOVIES data set to Excel.

PROC REPORT Code and ODS Excel Statement

```
ODS Excel FILE='PRINT-to-Excel.XLSX'  
          STYLE=BarrettsBlue ;  
PROC REPORT DATA=mydata.movies NOWINDOWS ;  
  COLUMNS category rating title length ;  
  DEFINE title    / DISPLAY 'Movie Title'  WIDTH=30 ;  
  DEFINE rating   / DISPLAY 'Movie Rating' WIDTH=5  
                  CENTER ;  
  DEFINE category / ORDER   'Category'     WIDTH=20 ;  
  DEFINE length   / DISPLAY 'Movie Length' WIDTH=3  ;  
RUN ;  
ODS Excel CLOSE ;
```

Excel Results

1	Category	Movie Rating	Movie Title	Movie Length
2	Action	PG-13	Jurassic Park	127
3	Action Adventure	R	Brave Heart	177
4		PG	Jaws	125
5		PG	Rocky	120
6		PG	The Hunt for Red October	135
7	Action Cops & Robber	R	Lethal Weapon	110
8		R	Scarface	170
9	Action Sci-Fi	PG	Star Wars	124
10		R	The Terminator	108
11	Adventure	G	The Wizard of Oz	101
12	Comedy	PG-13	Christmas Vacation	97
13		R	Coming to America	116
14		PG-13	National Lampoon's Vacation	98
15	Drama	PG	Casablanca	103
16		PG-13	Forrest Gump	142
17		PG-13	Michael	106
18	Drama Mysteries	R	Dressed to Kill	105
19	Drama Romance	PG-13	Ghost	127
20		PG-13	Titanic	194
21	Drama Suspense	R	Silence of the Lambs	118
22	Horror	R	Dracula	130
23		PG	Poltergeist	115

Conclusion

SAS users have many ways to create quality reports and output, including the use of DATA _NULL_ reporting techniques and an assortment of detail, summary, statistical, frequency, graphical and tabular procedures to choose from. But one method is typically chosen by today's SAS user more than any other approach – PROC REPORT. This paper illustrated only a few of the many capabilities that PROC REPORT offers. SAS users have the ability to create and deliver quality custom reports and specialized output using an assortment of powerful statements and options.

References

- Carpenter, Arthur L. (2016). "[Color, Rank, Count, Name; Controlling it all in PROC REPORT,](#)" Proceedings of the 2016 SAS Global Forum (SGF) Conference.
- Cochran, Ben (2005). "[A Gentle Introduction to the Powerful REPORT Procedure,](#)" Proceedings of the 2005 SAS Users Group International (SUGI) Conference.
- Lafler, Kirk Paul (2016). "[An Introduction to PROC REPORT,](#)" Proceedings of the 2016 SouthEast SAS Users Group (SESUG) Conference.
- Lafler, Kirk Paul (2016). "[An Introduction to PROC REPORT,](#)" Proceedings of the 2016 Nebraska SAS Users Group (NEBSUG) Conference.
- LeBouton, Kimberly (2014). "[Getting Up to Speed with PROC REPORT,](#)" Proceedings of the 2014 Western Users of SAS Software (WUSS) Conference.
- Lewandowski, David (2008). "[A Step-by-Step Introduction to PROC REPORT,](#)" Proceedings of the 2008 SAS Global Forum (SGF) Conference.
- Pass, Ray and Sandy McNeill (2006). "[PROC REPORT: Doin' It in Style,](#)" Proceedings of the 2006 SAS Users Group International (SUGI) Conference.

Acknowledgments

The author thanks Clarence Jackson and Greg Gengo, SouthCentral SAS Users Group (SCSUG) Conference Co-Chairs for accepting my abstract and paper; the SouthCentral SAS Users Group (SCSUG) Executive Board; and SAS Institute for organizing and supporting a great conference!

Trademark Citations

SAS and all other SAS Institute Inc. product or service names are registered trademarks or trademarks of SAS Institute Inc. in the USA and other countries. ® indicates USA registration. Other brand and product names are trademarks of their respective companies.

About the Author

Kirk Paul Lafler is an entrepreneur, consultant and founder of Software Intelligence Corporation, and has been using SAS since 1979. Kirk is a SAS application developer, programmer, certified professional, provider of IT consulting services, mentor, advisor, professor at UC San Diego Extension, educator to SAS users around the world, and emeritus sasCommunity.org Advisory Board member. As the author of six books including Google® Search Complete (Odyssey Press. 2014) and PROC SQL: Beyond the Basics Using SAS, Second Edition (SAS Press. 2013); Kirk has written hundreds of papers and articles; been an Invited speaker and trainer at hundreds of SAS International, regional, special-interest, local, and in-house user group conferences and meetings; and is the recipient of 25 “Best” contributed paper, hands-on workshop (HOW), and poster awards.

Comments and suggestions can be sent to:

Kirk Paul Lafler

SAS® Consultant, Application Developer, Programmer, Data Analyst, Educator and Author

Software Intelligence Corporation

E-mail: KirkLafler@cs.com

LinkedIn: <http://www.linkedin.com/in/KirkPaulLafler>

Twitter: @sasNerd